
Computer Lab 06 Eigenvalues

In science and engineering an important goal is to become a skilled prac-
titioner by doing it yourself. The computer labs provide computational
experience related to the analytic theory presented in the lectures.

The Eigenvalue Algorithm of Jacobi

This lab is about using an iterative technique involving plane rotations to
find the eigenvalues of a symmetric matrix.

When A ∈ Rn×n with A = AT the spectral theorem states the eigen-
values λi of A are real and the corresponding eigenvectors ξi may be chosen
to form an orthonormal basis of Rn. Thus A = QDQT where D is the
diagonal matrix and Q is the orthogonal matrix given by

D =


λ1 0 · · · 0
0 λ2 · · · 0
...

...
. . .

...
0 0 · · · λn

 and Q =

 ξ1

∣∣∣∣∣ ξ2
∣∣∣∣∣ · · ·

∣∣∣∣∣ ξn


Jacobi’s method for solving the eigenvalue-eigenvector problem con-
structs a sequence of plane rotations which in the limit reduce the symmet-
ric matrix A to a diagonal matrix. To understand how the method works
consider the symmetric matrix A ∈ R2 and the rotation R given by

A =

[
α β
β δ

]
and R =

[
cosφ sinφ
− sinφ cosφ

]
.

For convenience denote c = cosφ and s = sinφ. Now compute

RTAR =

[
c −s
s c

] [
α β
β δ

] [
c s
−s c

]
=

[
c −s
s c

] [
αc− βs αs+ βc
βc− δs βs+ δc

]
=

[
αc2 − βcs− βcs+ δs2 αcs+ βc2 − βs2 − δcs
αcs− βs2 + βc2 − δcs αs2 + βcs+ βcs+ δc2

]
=

[
αc2 − 2βcs+ δs2 (α− δ)cs+ β(c2 − s2)

(α− δ)cs+ β(c2 − s2) αs2 + 2βcs+ δc2

]
.

The above matrix will be diagonal provided c and s are chosen such that

(α− δ)cs+ β(c2 − s2) = 0.

To solve for c and s substitute t = s/c so that

cs = c2t and c2 − s2 = c2(1− t2).

Upon canceling c2 it follows that t satisfies

(α− δ)t+ β(1− t2) = 0.

To finish we need to show this equation has real solutions in t and then
express c and s in terms of t.

If β = 0 then the original matrix was diagonal and there is nothing to
do. Otherwise, β ̸= 0 and the equation is quadratic. In standard form we
have

−βt2 + (α− δ)t+ β = 0

with discriminant
(α− δ)2 + 4β2 ≥ 0.

Since the discriminant is non-negative, the solutions t are real.
Now solve for c and s as follows. By the Pythagorean theorem

1 = c2 + s2 = c2(1 + t2).

Therefore

c =
1√

1 + t2
and s =

t√
1 + t2

.

We remark that c ∈ [0, 1] and s ∈ [−1, 1] implies φ ∈ [−π/4, π/4].
To generalize the above idea to larger matrices A ∈ Rn×n, consider

the rotation in the plane spanned by the basis vectors ep and eq holding
the other directions constant. Namely, let R(pq)(φ) be the matrix whose
elements rij are the same as the identity except for the four elements

rpp = cosφ, rpq = sinφ
rqp = − sinφ, rqq = cosφ.

Now construct a sequence of matrices A(k) such that A(k) → D where
D is the diagonal matrix of eigenvalues of A. Let A(0) = A and define

A(k+1) = R(pq)(φk)
TA(k)R(pq)(φk)

where p and q are such that

|A(k)
pq | = max

{
|A(k)

ij | : i ̸= j
}

and φk is chosen as in the n = 2 case so A
(k+1)
pq = 0 with the additional

requirement to choose t to be smallest in absolute value from among the
two possible choices given by the quadratic equation.

The goal today is to write a program that uses the Jacobi method
to find the eigenvalues of an individualized matrix. Your matrix may be
obtained by clicking on the following link:

https://fractal.math.unr.edu/~ejolson/466-22/eigvals/matrix.cgi

Please do not use anyone else’s matrix for this lab.

Solving Quadratic Equations

As already seen, finding the values of cosφk and sinφk that appear in the
rotation R(pq)(φk) involve solving a quadratic equation. In the general case
that equation is

−apqt
2 + (app − aqq)t+ apq = 0

where aij are the entries of the matrix A(k).
We discussed solving quadratic equations in the introductory lab; how-

ever, in this case extra care needs to be taken to reduce rounding errors
because the size of the off-diagonal entries of A(k) becomes small in com-
parison to the diagonal entries as A(k) → D.

The usual quadratic formula gives

t =
−b±

√
b2 − 4ac

2a

where a = −apq, b = app − aqq and c = apq. The off-diagonal terms being
small means a and c are small. Thus,√

b2 − 4ac ≈ |b| and so − b+
√

b2 − 4ac ≈ 0 if b > 0.

To avoid the loss of precision which comes from the near cancellation
of two nearly equal numbers, rewrite the formula as

−b+
√
b2 − 4ac

2a
· b+

√
b2 − 4ac

b+
√
b2 − 4ac

=
−2c

b+
√
b2 − 4ac

.

Now b > 0 with a and c small implies

b+
√

b2 − 4ac ≈ b+ |b| ≈ 2b

and there is no loss of precision. If b < 0 a similar technique must be used
to find the solution corresponding to the negative square root.

Following is a Julia function that employs the above ideas to find the
solutions to the quadratic equation at2 + bt+ c = 0.

1 function quadsolv(a,b,c)
2 sq=sqrt(b^2-4*a*c)
3 np=-b+sq
4 nm=-b-sq
5 if abs(nm)<abs(np)
6 t1=np/(2*a)
7 t2=2*c/np
8 else
9 t1=2*c/nm

10 t2=nm/(2*a)
11 end
12 return t1,t2
13 end

To test how it is working try solving

t2 + 10000t+ 1 = 0

and compare the result to the solution computed using the usual formula.

julia> a=1; b=10000; c=1
1

julia> p(t)=a*t^2+b*t+c
p (generic function with 1 method)

julia> t1,t2=quadsolv(a,b,c)
(-0.00010000000100000001, -9999.999899999999)

julia> tbad=(-b+sqrt(b^2-4*a*c))/(2*c)
-0.00010000000111176632

julia> p(t1)
0.0

julia> p(tbad)
-1.1176630732023796e-9

Note 8 digits are different between t1 computed using the transformed
expression versus tbad obtained from the usual quadratic formula.

Coding the Jacobi Eigenvalue Solver

The following discussion concerns the matrix which appears when I click
the above link. That matrix is different than what you will obtain when
you click the same link. To finish this lab please repeat these same steps
but for own individualized matrix.

Upon clicking on the link, I obtained

Start editing a new file called lab06.jl and insert using LinearAlgebra at
the top. Then copy the matrix with your mouse and paste it below. The
beginning of your program should now look similar to

1 using LinearAlgebra
2

3 A=[ 8.52 -1.80 1.08 -3.27 -3.25;
4 -1.80 1.76 -6.67 -2.67 4.05;
5 1.08 -6.67 1.16 -8.34 2.78;
6 -3.27 -2.67 -8.34 7.18 6.27;
7 -3.25 4.05 2.78 6.27 -3.28 ]

After adding the quadratic solver developed in the previous section
write a new function getpq that find p and q such that

|Apq| = max
{
|Aij | : i ̸= j

}
.

One way to find p and q is with the built-in maximum operator followed by
findfirst applied over the Cartesian indices such that i ̸= j. I found the
functional approach too confusing and decided to use a nested loop instead.

23 function getpq(A)
24 N,_=size(A)
25 m=A[1,2]; p=1; q=2
26 for i=1:N-1
27 for j=i+1:N
28 if abs(A[i,j])>m
29 m=abs(A[i,j]); p=i; q=j
30 end
31 end
32 end
33 return p,q
34 end

One advantage of the Julia just-in-time compiler over interpreted languages
is that loops such as written above perform well.

What remains is to write a jacobi function that computes the elements
bij of R(pq)(φk)

TA(k)R(pq)(φk). As computed for the case n = 2 we have

bpp = appc
2 − 2apqsc+ aqqs

2

bqq = apps
2 + 2apqsc+ aqqc

2

bpq = 0

bqp = 0.

A similar computation—not shown—provides the off diagonal entries along
the p and q columns and rows as

bpi = bip = aipc− aiqs

bqi = biq = aiqs+ aiqc
for i ̸= p and j ̸= q.

Since the rest of the matrix R is given by the identity, the rest of the
elements are unchanged. Here is a possible implementation.

36 function jacobi(A)
37 p,q=getpq(A)
38 t1,t2=quadsolv(-A[p,q],A[p,p]-A[q,q],A[p,q])
39 t=t1
40 if abs(t2)<abs(t1)
41 t=t2
42 end
43 c=1/sqrt(1+t^2)
44 s=c*t
45 B=copy(A)
46 B[p,p]=A[p,p]*c^2-2*A[p,q]*s*c+A[q,q]*s^2
47 B[q,q]=A[p,p]*s^2+2*A[p,q]*s*c+A[q,q]*c^2
48 B[p,q]=0
49 B[q,p]=0
50 N,_=size(A)
51 for i=1:N
52 if i==p||i==q
53 continue
54 end
55 B[i,p]=A[i,p]*c-A[i,q]*s
56 B[p,i]=B[i,p]
57 B[i,q]=A[i,p]*s+A[i,q]*c
58 B[q,i]=B[i,q]
59 end
60 return B
61 end

Note that line 37 calls getpq to determine which off-diagonal element
of A has the largest absolute magnitude so it can be rotated to zero. Lines
39 through 42 choose the smallest t from among the two roots which
solve the quadratic equation. Finally, lines 46 through 59 compute B =
R(pq)(φk)

TA(k)R(pq)(φk) using the relations just discussed.

To complete this assignment write a loop that performs 30 iterations
using a statement like

Ak=jacobi(Ak)

inside a loop. Then print out the approximately-diagonal matrix A(30). The
output should like like

A(30)=
5×5 Matrix{Float64}:
7.12263 7.22391e-23 -1.809e-16 0.0 -1.56938e-33
7.22391e-23 5.98028 0.0 -4.57679e-17 -8.3952e-13
-1.809e-16 0.0 -14.0027 -4.90817e-28 2.59551e-14
0.0 -4.57679e-17 -4.90817e-28 16.6483 1.04593e-21
-1.56938e-33 -8.3952e-13 2.59551e-14 1.04593e-21 -0.408503

For extra credit extract the eigenvalue approximations from the diago-
nal and sort them with sort(diag(Ak)). Then compare your approximation
to the output of eigvals(A) which is part of the LinearAlgebra library. De-
termine which approximation is more accurate by plugging the eigenvalues
into the characteristic polynomial.

Submitting Your Work

One pdf file should be submitted for grading that contains two parts:

• A program that performs 30 iterations of the Jacobi algorithm for
finding eigenvalues and produces output similar to above for your in-
dividualized matrix.

• The output from running that program.

After debugging and making sure your program runs correctly, please
prepare your submission by typing

$ julia lab06.jl >lab06.out
$ j2pdf -o lab06.pdf lab06.jl lab06.out

Before uploading, check lab06.pdf with

$ evince lab06.pdf &

to make sure the program and output looks correct. Please reboot into
Microsoft Windows before leaving the lab.

using LinearAlgebra

 A=[  8.52 -1.80  1.08 -3.27 -3.25;
     -1.80  1.76 -6.67 -2.67  4.05;
      1.08 -6.67  1.16 -8.34  2.78;
     -3.27 -2.67 -8.34  7.18  6.27;
     -3.25  4.05  2.78  6.27 -3.28 ]

function quadsolv(a,b,c)
    sq=sqrt(b^2-4*a*c)
    np=-b+sq
    nm=-b-sq
    if abs(nm)<abs(np)
        t1=np/(2*a)
        t2=2*c/np
    else
        t1=2*c/nm
        t2=nm/(2*a)
    end
    return t1,t2
end

function getpq(A)
    N,_=size(A)
    m=A[1,2]; p=1; q=2
    for i=1:N-1
        for j=i+1:N
            if abs(A[i,j])>m
                m=abs(A[i,j]); p=i; q=j
            end
        end
    end
    return p,q
end

function jacobi(A)
    p,q=getpq(A)
    t1,t2=quadsolv(-A[p,q],A[p,p]-A[q,q],A[p,q])
    t=t1
    if abs(t2)<abs(t1)
        t=t2
    end
    c=1/sqrt(1+t^2)
    s=c*t
    B=copy(A)
    B[p,p]=A[p,p]*c^2-2*A[p,q]*s*c+A[q,q]*s^2
    B[q,q]=A[p,p]*s^2+2*A[p,q]*s*c+A[q,q]*c^2
    B[p,q]=0
    B[q,p]=0
    N,_=size(A)
    for i=1:N
        if i==p||i==q
            continue
        end
        B[i,p]=A[i,p]*c-A[i,q]*s
        B[p,i]=B[i,p]
        B[i,q]=A[i,p]*s+A[i,q]*c
        B[q,i]=B[i,q]
    end
    return B
end

Ak=copy(A)
for k=1:30
    global Ak=jacobi(Ak)
end

println("A(30)=")
display(Ak)


